```
/* aml to draw multiple graphs on a single page
/* Michael Silberbauer Jan 1994
/* Michael Silberbauer modified October 1994 to allow automatic
/* printing of entire drainage regions for macro or trace variables.
/* Michael Silberbauer modified November 1994 to print a record file and
/* to try to get duplex printing on the HP4Si laserjet (no such luck!)
/* All duplex commands commented out using /*dup
/* Modified to use point coverage of stations to select stations to plot:
/* Michael Silberbauer Feb 1995
/* test version MJS Mar 1995 (different inorg.dat, flow incorporated)
/* modified for Arc/Info 7.0.3 and Solaris 2.4 E Vermaak & M Silberbauer jul95
/* shading modified and renamed barcode.aml M Silberbauer Feb 1996
/* Large fonts option added Sep 1996, M Silberbauer
/* Version 2.0 with new font and shading Nov 1996, M Silberbauer
/* Version 2.1 with quaternary mini-map, Feb 1997, Michael Silberbauer
/* Version 2.2 with bug removal:
/*             a) values set to null before STATISTICS
/*             b) rounding of results for y-axis improved
/*                                     Jun 1997, Michael Silberbauer
/*             c) rounding changed because of numeric overflow in 7.1.1
/*                                     Aug 1997, Michael Silberbauer
/*             d) ISO dates            Nov 1998, Michael Silberbauer
/*             e) HP8000 printer       Jun 1999, Michael Silberbauer
/* Version 3.0 with median, number     Jun 1999, Michael Silberbauer
/* Version 3.1 with percentiles        Sep 2000, Michael Silberbauer
/* Version 3.2 with Arc/Info 8 fixes
/*  menu improvements, log file        Jan 2001, Michael Silberbauer
/* Version 3.3 percentile bug fix      May 2001, Michael Silberbauer
/* Version 3.31 remove all temp files  Jun 2001, Michael Silberbauer
/* Version 3.4 allow variable file name  Jun 2001, Michael Silberbauer
/* Version 3.41 fix variable file name!  Oct 2001, Michael Silberbauer
/* Version 3.5  adapt to Windows 2000  Apr 2002, Michael Silberbauer
/* Version 3.51 add PDF conversion     Jun 2002, Michael Silberbauer

/* Be sure to change the version number in the AML below!

/* Input specifications on command line:

&args CatPri CatSec CatTer Station VarType Flow Year1 Year2 Display Pcntl BigFont StnType noZ inFile Type Debug
```

```
&sv Version = %AML$FULLFILE% v3.51

&sv opsys = [extract 1 [show &os] ]

&if %opsys% = SunOS      &then
&do
  &sv enterprise = /hri
  &sv chemdir = %enterprise%/db/hri2/waterm/wmdata/
  &sv covdir  = %enterprise%/db/hri/cover/
  &sv tempdir = %enterprise%/tmp
  &sv pdfdir  = %tempdir%/
  &sv NL = \
  &sv remove = \rm
&end
&if %opsys% = Windows_NT &then
&do
  &sv enterprise = N:
  &sv chemdir = c:/temp/avtemp/
  &sv covdir  = C:/Data_large/av/
  &sv tempdir = C:/tmp
  &sv pdfdir  = C:\temp\pdf\in\
  &sv NL = /&
  &sv remove = del
  &sv Version = [unquote [subst [quote %Version%] '\' '\\']]
&end

&type ********************* Command-line interface! *********************
&type %opsys% operating system using directories %chemdir% and %covdir%
&call ISOdate
&type Running %Version% on %ISOdate%
&if [null %CatPri%] or [quote %CatPri%] = #    and ~
     ( [quote %CatSec%] <> # or [quote %CatTer%]  <> # ) and ~
      [quote %Station%] <> #  &then
&do
 &type Usage: barcode CatPri CatSec CatTer Station VarType Flow Year1 Year2 Display
 &type      Pcntl BigFont StnType noZ inFile Type Debug
 &type CatPri    limits the search to one primary catchment, (e.g. C, or # to skip)
 &type CatSec    narrows the search to a secondary catchment, if CatPri is set
 &type CatTer    further narrows the search if CatPri and CatSec are set
```

```
 &type Station   selects only one station, or CatQat
 &type VarType   macro or trace,
 &type Flow      to plot flow (where available)
 &type Year1     the first year to plot, e.g. 1983
 &type Year2     the last  year to plot, e.g. 1994
 &type Disp      display is p to plot to printer (s to screen or e to eps)
 &type Pcntl     percentile (= 90 percent by default)
 &type BigFont   for setting large fonts for posters and transparencies
 &type StnType   R or H (default is B for both)
 &type noZ       Z or noZ (to exclude Z stations)
 &type inFile    input data file (ignore)
 &type Type      sun1 sun2 sun3 sun4 tek1040 tek4107 (ignore)
 &type Debug     used by programmers to set debugging mode (ignore)
 &type
 &type e.g. for single station : barcode # # # c9r002q01 macro flow 1980 1996 print
 &type e.g. for quat  catchment: barcode A 2 1 H macro flow 1993 1994 print
 &type e.g. for Pcntl, big font: barcode # # # A2H027Q01 macro flow 1993 1994 p 50 B
 &type e.g. for Debugging      : barcode E # # # m f 1991 2000 p 95 s b z # # d
 &type WARNING: DO NOT RUN MULTIPLE SESSIONS IN THE SAME DIRECTORY!
  &return

&end

&if ^ [exists . -workspace] &then createworkspace .
&if [null %CatSec%] &then &sv CatSec = #
&if [null %CatTer%] &then &sv CatTer = #
&if [null %Station% ] &then &sv Station  = #
&sv CatPri  = [translate %CatPri%]
&sv CatSec  = [translate %CatSec% ]
&sv CatTer  = [translate %CatTer% ]
&sv Station = [translate %Station%  ]
  &sv Single = .FALSE.
&if %Station% <> # and [length [quote %Station%]] > 1 &then
  &sv Single = .TRUE.
&if %Station% = # or [length [quote %Station%]] = 1 &then
  &sv CatQat = %Station%
&if [null %VarType% ] or [quote %VarType%] = '#' &then &sv VarType = Macro
&sv VarType = [substr [locase %VarType%] 1 1]
&sv Flow    = [substr [locase %Flow%] 1 1]
```

```
&if [null %Year1%] &then &sv Year1 = 1983
&if [null %Year2%] &then &sv Year2 = 1998
&if [null %Display%] &then &sv Display = p
  &else &sv Display = [substr %Display% 1 1]
&sv Display = [locase %Display%]
&if [null %Pcntl%] or [quote %Pcntl%] = '#' &then &sv Percentile = 90
  &else &sv Percentile = %Pcntl%
/* The following is a relic from the Tektronix days - it remains for sentimental reasons:
&if [null %Type%]    &then
&do
  &sv MoniType = '9999 3'
  &sv Type sun3
&end
&else
 &do
   &sv Type = [locase %Type%]
   &sv MoniType = '9999 3'
   &if %Type% = tek1040 &then &sv MoniType = 'tek1040'
   &if %Type% = tek4107 &then &sv MoniType = 'tek4107'
   &if %Type% = sun1    &then &sv MoniType = '9999'
   &if %Type% = sun2    &then &sv MoniType = '9999 2'
   &if %Type% = sun3    &then &sv MoniType = '9999 3'
   &if %Type% = sun4    &then &sv MoniType = '9999 4'
 &end
&if %Display% = p &then
&do
  &sv MoniType = 'printer'
  &sv Type     = paper
&end

&if %Display% = e &then
&do
  &sv MoniType = 'printer'
  &sv Type     = postscript
&end

&if [substr [locase %BigFont%] 1 1] = 'b' &then &sv BigFont = .true.
&else &sv BigFont = .false.
```

```
&if [null %StnType%] &then &sv StnType = B
&else &sv StnType = [translate %StnType%]

&if [null %noZ%] &then &sv noZ = NOZ
&else &sv noZ = [translate %noZ%]
&sv no_zstn = .false.
&if %noZ% = NOZ &then &sv no_zstn = .true.

&if [null %inFile%] or %inFile% = # &then &sv inFile = Default

&if [substr [locase %Debug%] 1 1] = 'd' &then &sv Debug = .true.
&else &sv Debug = .false.

&call SetDefaults
&sys echo User %User% running %Version% on %ISOdate% >> %LogFile%
&call CheckDate
&sv rw = [write %ReportUnit% [quote %ISOdate1% to %ISOdate2% on IWQS GIS file %ChemFile%]]

&type using: barcode %CatPri% %CatSec% %CatTer% %Station% %VarType% %Flow% %Year1% %Year2% %Display% %Type%
BigFont=%BigFont% StnType=%StnType% noZ=%noZ% Debug=%Debug%
&sys echo barcode %CatPri% %CatSec% %CatTer% %Station% %VarType% %Flow% %Year1% %Year2% %Display% %Type%
BigFont=%BigFont% StnType=%StnType% noZ=%noZ% Debug=%Debug% ChemFile=%ChemFile%>> %LogFile%
&type ************************************************************

ARCPLOT

&type Searching for valid stations...
clearselect %StnPnt% point
&if %Single% &then reselect %StnPnt% point station = [quote %Station%]
&else
  &do
   reselect %StnPnt% point primary = [quote %CatPri%]
   &if [quote %CatSec%] <> # &then
     reselect %StnPnt% point secondary = [quote %CatPri%%CatSec%]
   &if [quote %CatTer%] <> # &then
     reselect %StnPnt% point tertiary  = [quote %CatPri%%CatSec%%CatTer%]
   &if [quote %CatQat%] <> # &then
     reselect %StnPnt% point quaternary = [quote %CatPri%%CatSec%%CatTer%%CatQat%]
   &if %StnType% <> B &then
```

```
     reselect %StnPnt% point stntype = [quote %StnType%]

   reselect %StnPnt% point monitype = 'Q'

  &end

&sv nStations = [before [show select %StnPnt% point] ,]
infofile %StnPnt% point %BarTmpLst% primary secondary tertiary quaternary station init
&data ARC INFO
  ARC
  SELECT [translate [entryname %BarTmpLst%]]
  SORT ON QUATERNARY, STATION
  Q STOP
&end

list %BarTmpLst% info 1 10

&type Number of stations selected = %nStations%
clearselect %BarTmpLst% info
reselect    %BarTmpLst% info station <> ''

&if %nStations% = 0 &then
&do
  QUIT
  &stop No stations found...
&end

/*&message &off

&do n = 1 &to %nStations%
   &if [exists %BarTmpXY%   -file] &then &ty [delete %BarTmpXY%   -file]
   &if [exists %BarTmpGeo%  -file] &then &ty [delete %BarTmpGeo%  -file]
   &if [exists %BarPrjXY%   -file] &then &ty [delete %BarPrjXY%   -file]
   &if [exists %BarPrjGeo%  -file] &then &ty [delete %BarPrjGeo%  -file]
   &if [exists %BarScratch% -file] &then &ty [delete %BarScratch% -file]
   &if [exists %BarPrjScr% -file]  &then &ty [delete %BarPrjScr% -file]

   &if [show program] <> ARCPLOT &then ARCPLOT
```

```
&sv Station     = [show select %BarTmpLst% info %n% item station]

&if [substr [quote %Station%] 1 1] = 'Z' and %no_zstn% &then
  &type (%n%) Skipping Z-station %Station%
&else
&do

  clearselect %StnPnt% point
  reselect    %StnPnt% point station = [quote %Station%]

  &if [iteminfo %StnPnt% -point place -exists] &then
    &sv Place     = [show select %StnPnt% point 1 item place]
  &else
    &sv Place     = [unquote '']

  &sv Description = [show select %StnPnt% point 1 item description]
  &sv mx          = [show select %StnPnt% point 1 item x-coord]
  &sv my          = [show select %StnPnt% point 1 item y-coord]
  &sv Primary     = [show select %StnPnt% point 1 item primary]
  &sv Secondary   = [show select %StnPnt% point 1 item secondary]
  &sv Tertiary    = [show select %StnPnt% point 1 item tertiary]
  &sv Quaternary  = [show select %StnPnt% point 1 item quaternary]

  &sv PrjUnit     = [open %BarTmpXY% openstatus -write]
  &sv wp [write %PrjUnit% [quote %mx% %my%]]
  &sv cl [close %PrjUnit%]
  &call AlbGeo

  &sv Place       [unquote [trim [quote %Place%]]]
  &sv Description [unquote [trim [quote %Description%]]]
  &sv Description %Description% - %Place%
  &type (%n% of %nStations%) %Station% %GeoString% - %Description%
  &sv PlotName    = %Station%

  &sv nVars  = %nVariables%
  &if %Flow% = f &then &call FlowCalc

  &call GraphLoop
```

```
     &if [show program] = ARCPLOT &then QUIT /* ARCPLOT (so as to close .gra file)

     &if %Display% = p &then &call PaperPrint
     &if %Display% = e &then &call PostScriptPrint
     &if %Display% = s &then &call ScreenPrint
   &end
&end

&message &on

&sv CloseReport = [close -all]
&if %opsys% = SunOS &then &sys unix2dos %Report% %Report%

/* The report file has become too wide to print, even with landscape, so
/* generate a comma-delimited file for use with spreadsheets (MJS 2000).

&sys echo Results are summarised in file %Report% >> %LogFile%
&type Results are summarised in file %Report%
&type Combine files with a command of the form:
&type grep -v -h file *txt | grep -v Data | grep -v ":" > all.txt
&call ISOdate
&type End of station list at %ISOdate%

&if [show program] = ARCPLOT &then QUIT /* arcplot session

&if ^ %Debug% &then &call DeleteTempFiles
&return

/* - - - - - - - - - - - - - - - - - -

&routine GraphLoop
/* routine to draw graphs:

&sv counter = %counter% + 1

/*&if %counter% > 2 &then &stop Finished - test mode.

&if ( %Display% = p ) OR ( %Display% = e ) &then
&do
```

```
  display 1040
  %PlotName%
  &type ...creating graphics file %PlotName%
&end
  &else display [unquote %MoniType%]

LIBRARY access
PAGEUNITS cm
PAGESIZE %xPageSiz% %yPageSiz%
BOX 0 0 %xPageSiz% %yPageSiz%
BOX %xMar% %yMar% %xPage% %yPage%
&sv yMin = 0
&sv yMax = %nVars%
&sv yMap = 0.26
GRAPHEXTENT %.startday% %yMin% %.endday% %yMax%
GRAPHLIMITS %Lxmin% %Lymin% %Lxmax% %Lymax%
BOX         %Lxmin% %Lymin% %Lxmax% %Lymax%
&sv xLencm  = [calc %Lxmax%   - %Lxmin%]
&sv yLencm  = [calc %Lymax%   - %Lymin%]
&sv xLenDay = [calc %.endday% - %.startday%]
&sv yLenVar = [calc %yMax%    - %yMin%]
&sv cm_Day  = [calc %xLencm% / %xLenDay%]
&sv cm_Var  = [calc %yLencm% / %yLenVar%]
&sv tsm     = 14   /* textsymbol
&sv tss     = 22   /* textsymbol (special characters)
&sv tsi     = 23   /* textsymbol (italics)
&sv tsb     = 1    /* textsymbol ("big font")

UNITS page
TEXTSET font
&if %MoniType% = printer &then TEXTSET font
&else
&do
  &if [substr %MoniType% 1 1 ] = 9 &then TEXTSET font
  &if [substr %MoniType% 1 1 ] = 1 &then TEXTSET plotter
&end

&if %BigFont% &then
&do
```

```
  &sv tsm = %tsb%
  &sv tss = %tsb%
&end

TEXTSYMBOL %tsm%
TEXTSTYLE TYPESET
TEXTSCALE 2.5
&if %BigFont% &then TEXTSCALE 3.7
MOVE [calc %xPage% * 0.5] [calc %yPage% - 0.7]
TEXT [quote %Station%:  BARCODE GRAPHS OF WATER QUALITY VARIABLES] cc
TEXTSCALE 2
&if %BigFont% &then TEXTSCALE 2.6
MOVE [calc %xPage% * 0.5] [calc %yPage% - 1.4]
TEXTJUSTIFICATION cc
TEXT [quote %ISOdate1% to %ISOdate2% on IWQS INFO %ChemFile% cover %StnPnt%]

MOVE [calc %xPage% * 0.5] [calc %yPage% - 1.9]
TEXT [quote %Description% [unquote %GeoLabel%]]
&sv closeall [close -all]
&sv ReportUnit  = [open %Report% openstatus -append]
&sv rw = [write %ReportUnit% [quote %Station% %Quaternary%: %Description%]]

TEXTSYMBOL %tss%
TEXTSTYLE TYPESET
TEXTJUSTIFICATION cr
&if ^ %BigFont% &then
&do
  move [calc %xPage% * 0.97] [calc %yPage% * 0.97]
  TEXT [quote (%Percentile%%suffix% Pcntl) Maximum mg[unquote '!pat7248;!pat7318;!pat1619;']\minimum (median)
values] cr
&end

&sv x = %Lxmin%
TEXTJUSTIFICATION cl
SHADESET colornames

UNITS graph
AXIS vertical
/*AXISHATCH 1 0
```

```
AXIS vertical %.endday% 0
TEXTSYMBOL %tsm%
TEXTSTYLE TYPESET
TEXTOFFSET 0 0
/*TEXTOFFSET 1.5 0
/*AXISHATCH 1 0
TEXTJUSTIFICATION cr
AXIS horizontal
&if %BigFont% &then
&do
  &sv tsc [show textscale]
  TEXTSCALE 0.4
  AXISDATE M YY NoMonth NoDay
  textscale %tsc%
&end
&else
  AXISDATE M Year NoMonth NoDay

/*&if %Debug% &then &echo &on

TEXTJUSTIFICATION cl
TEXTSCALE 1.5
SHADESET colornames
&sv NoData_AtAll = .TRUE.
&sv ValList = %Station%
&type Drawing bar graphs...

&sv VarList = Station

&do n = 1 &to %nVars%
   /* Label the left-hand y-axis with the constituents:
   &sv y = [calc %Lymin% + [calc ( %n% - 0.5 ) * %cm_Var% ]]
   MOVE [calc %x% - 0.1] %y%

   &if %BigFont% &then
   &do
     &sv tsc [show textscale]
     TEXTSCALE 2.5
     TEXT [value VarAbbr%n%] ur
```

```
    TEXTSCALE %tsc%
  &end
  &else
    TEXT [value Var%n%] lr

  &sv VarList = %VarList%,[value Var%n%]min,[value Var%n%]med,[value Var%n%]p%Percentile%,[value Var%n%]max
  &sv y1 = [calc %Lymin% + [calc %n% -   1] * %cm_Var%]
  &sv y2 = [calc %Lymin% + [calc %n% - 0.1] * %cm_Var%]
  GRAPHLIMITS %Lxmin% %y1% %Lxmax% %y2%
  linecolor 6
  line %Lxmin% %y2% %Lxmax% %y2%
  linecolor 1

  &if ( %flow% = f and %n% = %nVars% ) &then &call PlotFlow
  &else
    &call PlotChem
&end

&if %counter% = 1 &then
  &sv rw = [write %ReportUnit% [quote %VarList%]]

&sv rw = [write %ReportUnit% [quote %ValList%]]

&type Drawing the location maps...

MAPPOSITION cen cen
LINESYMBOL 1
markerset mineral
&call pri_location
&call sec_location
&call ter_location
&call qat_location
&call datestamp

/*&if %Display% = s &then &pause
&if %Display% = s &then &call ScreenSaver

/*&if %Debug% &then &echo &off
```

```
&return

/* - - - - - - - - - - - - - - - - - -

&routine datestamp
UNITS page
&sv x1 = [calc 0.75 * %xPage%]
&sv y1 = %yMar% + 0.1
&sv x2 = [calc %xPage% - 0.1 ]
&sv y2 = [calc %yMap% * %yPageSiz%]
/*BOX %x1% %y1% %x2% %y2%
move [calc %xPage% - %xMar% ] %yMar%
textstyle typeset
textscale 1.4
textangle 90

&call ISOdate
&sv Address = Institute for Water Quality Studies\Department of Water Affairs & Forestry\Private Bag X313
PRETORIA.\tel (012) 808 0375, fax (012) 808 2702\E-mail: SilberbauerM@dwaf.gov.za
&sv stntag %Station% - %counter%-
TEXT [quote %stntag%\%Version%\%Address%\Run %ISOdate% (by %User%)\Text in %Report%] ll
maplimits %xMar% [calc %yPage% - ( 4 * %yMar% )] ~
          [calc 4 * %xMar% ] [calc %yPage% - %yMar%]
mape image %enterprise%/db/clip/rsa_logo.jpg
image %enterprise%/db/clip/rsa_logo.jpg composite 1 2 3
move [calc %x2% - 0.1] [calc %y1% + 0.1]
textangle 0
&return

/* - - - - - - - - - - - - - - - - - -

&routine pri_location
&type Drawing primary map:
units page
&sv mapwidth = [calc 0.21 * %xPageSiz%]
&sv xpri1 = [calc %xMar% + 0.1]
&sv ypri1 = [calc %yMar% + 0.1]
&sv xpri2 = [calc %xpri1% + %mapwidth%]
&sv ypri2 = [calc %yMap% * %yPageSiz%]
```

```
maplimits %xpri1% %ypri1% %xpri2% %ypri2%
mapextent -759000,-3853000,942000,-2316000
clipmapextent off
clearselect  .ponet poly
reselect     .ponet poly popytype = 2
polygonshade .ponet 57
clearselect  %province% poly
reselect     %province% poly class = 'Country' and name <> 'LESOTHO' and name <> 'SWAZILAND'
polygonshade %province% 32
ARCS         .ponet
CLEARSELECT  %Catch_Pri% poly
RESELECT     %Catch_Pri% poly primary = [quote %Primary%]
nselect      %Catch_Pri% poly
polygonsHADE %Catch_Pri% 32
polygons     %Catch_Pri%
CLEARSELECT  %Catch_Pri% poly
RESELECT     %Catch_Pri% poly primary = [quote %Primary%]
clearselect  %Catch_Sec% poly
reselect     %Catch_Sec% poly primary = [quote %Primary%]
linesymbol   1
polygons     %Catch_Sec%

UNITS map
markersymbol %MonPtMrk%
markerscale 1.1
markercolor 7
marker %mx% %my%
markerscale 0.9
markercolor 4
marker %mx% %my%
UNITS page
MOVE [extract 3 [show maplimits]] [extract 2 [show maplimits]]
textoffset -0.1 0.1
textmask rectangle 0.02

&if %BigFont% &then
&do
  &sv tsc [show textscale]
  TEXTSCALE 2.5
```

```
  TEXT [quote Drainage\region [substr %Station% 1 1]] lr
  TEXTSCALE %tsc%
&end
&else
  TEXT [quote Drainage\region [substr %Station% 1 1]] lr

BOX       [show maplimits]
textoffset 0 0
textmask none
CLIPMAPEXTENT on
&return

/* - - - - - - - - - - - - - - - - - -

&routine sec_location
&type Drawing secondary map:
units page
&sv xsec1 = [calc %xpri2% + 0.1]
&sv ysec1 = %ypri1%
&sv xsec2 = [calc %xsec1% + %mapwidth%]
&sv ysec2 = %ypri2%
maplimits %xsec1% %ysec1% %xsec2% %ysec2%
clearselect  %Catch_Pri% poly
reselect     %Catch_Pri% poly primary = [quote [substr %Station% 1 1]]
clearselect  %Catch_Sec% poly
reselect     %Catch_Sec% poly secondary = [quote %Secondary%]
mapextent poly %Catch_Sec%
clipmapextent off
clearselect  .ponet poly
reselect     .ponet poly popytype = 2
polygonshade .ponet 57
clearselect  %province% poly
reselect     %province% poly class = 'Country' and name <> 'LESOTHO' and name <> 'SWAZILAND'
polygonshade %province% 32
arcs         .ponet
nselect      %Catch_Sec% poly
polygonshade %Catch_Sec% 33
nselect      %Catch_Pri% poly
polygonshade %Catch_Pri% 32
```

```
clearselect  %Catch_Ter% poly
reselect     %Catch_Ter% poly secondary = [quote %Secondary%]
arcs         %Catch_Ter%
markersymbol %MonPtMrk%
markercolor 7
markerscale 1.85
units  map
marker %mx% %my%
markerscale 1.7
markercolor 4
marker %mx% %my%
UNITS  page
MOVE [extract 3 [show maplimits]] [extract 2 [show maplimits]]
textoffset -0.1 0.1
textmask rectangle 0.02

&if %BigFont% &then
&do
  &sv tsc [show textscale]
  TEXTSCALE 2.8
  TEXT [quote [substr %Station% 1 2]] lr
  TEXTSCALE %tsc%
&end
&else
  TEXT [quote [substr %Station% 1 2]] lr

BOX       [show maplimits]
textoffset 0 0
textmask none
CLIPMAPEXTENT on
&return

/* - - - - - - - - - - - - - - - - - -

&routine ter_location
&type Drawing tertiary map:
units page
&sv xter1 = [calc %xsec2% + 0.1]
&sv yter1 = %ypri1%
```

```
&sv xter2 = [calc %xter1% + %mapwidth%]
&sv yter2 = %ypri2%
maplimits %xter1% %yter1% %xter2% %yter2%
clearselect %Catch_Ter% poly
reselect %Catch_Ter% poly tertiary = [quote %Tertiary%]
MAPEXTENT poly %Catch_Ter%
CLIPMAPEXTENT off
&call SearchBox
clearselect  .ponet poly
reselect     .ponet poly popytype = 2
polygonshade .ponet 57
clearselect  %province% poly
reselect     %province% poly class = 'Country' and name <> 'LESOTHO' and name <> 'SWAZILAND'
polygonshade %province% 32
nselect      %Catch_Ter% poly
polygonshade %Catch_Ter% 33
clearselect  %Catch_Ter% poly
/*
/*ARCS %Catch_Ter%
clearselect   %Towns% poly
polygonshade  %Towns% 83
linesymbol 5
arcs %Catch_Qat%
RESELECT %Catch_Ter% poly tertiary = [quote %Tertiary%]
LINESYMBOL 14
polygons %Catch_Ter%
linesymbol    4
linecolor     cyan_2
arcs          %RiversOut%

&do nOrder = 1 &to 7
  &sv width = %nOrder% / 140
  &type Rivers of order %nOrder% being drawn with a width of ~
        [calc [round [calc 100 * %width%]] / 100]cm:
  clearselect %Rivers% arc
  reselect    %Rivers% arc order = %nOrder%
  linesize    %width%
  arcs        %Rivers%
&end
```

```
clearselect   %Lakes% poly
reselect      %Lakes% poly name <> 'Island'
polygonshade  %Lakes% 56

overpost      0.1 1.0 0.1
overpost text moveable
overpost on
textalignment automatic

LINESYMBOL 7
ARCS .ponet
LINESYMBOL 1
markerSYMBOL %MonPtMrk%
markerSCALE 2.5
markerCOLOR 7
UNITS  map
marker %mx% %my%
markerSCALE 2.3
markercolor 4
marker %mx% %my%
UNITS  page

reselect %Towns% poly mapextent
reselect %Towns% poly overlap %Catch_Ter% poly
infofile %Towns% poly %BarTmpSrt% area name init
reselect %BarTmpSrt% info name <> ''
&sv nTowns = [extract 1 [show select %BarTmpSrt% info]]
&if %nTowns% > 0 &then
&do

  &data ARC INFO
    ARC
    SELECT [translate [entryname %BarTmpSrt%]]
    SORT ON AREA (D)
    Q STOP
  &end

  list %BarTmpSrt% info 1 6
```

```
  &sv BigTown = [show select %BarTmpSrt% info 1 item name]
  &sv BigArea = [show select %BarTmpSrt% info 1 item area]
  reselect      %Towns% poly name = [quote %BigTown%]
  &sv AreaMax = [calc %BigArea% + ( %BigArea% * 0.000001 )]
  &sv AreaMin = [calc %BigArea% - ( %BigArea% * 0.000001 )]
  reselect      %Towns% poly area > %AreaMin% and area < %AreaMax%
  labeltext     %Towns% nametag cc
&end

overpost off
MOVE [extract 3 [show maplimits]] [extract 2 [show maplimits]]
textoffset -0.1 0.1
textmask rectangle 0.02
&if %BigFont% &then
&do
  &sv tsc [show textscale]
  TEXTSCALE 2.8
  text %Tertiary% lr
  TEXTSCALE %tsc%
&end
&else
  text %Tertiary% lr

BOX       [show maplimits]
textoffset 0 0
textmask none
markerSCALE 1
CLIPMAPEXTENT on
&return

/* - - - - - - - - - - - - - - - - - -

&routine qat_location
&type Drawing quaternary map:
units page
&sv xqat1 = [calc %xter2% + 0.1]
&sv yqat1 = %ypri1%
&sv xqat2 = [calc %xqat1% + %mapwidth%]
&sv yqat2 = %ypri2%
```

```
maplimits %xqat1% %yqat1% %xqat2% %yqat2%
clearselect %Catch_Qat% poly
reselect    %Catch_Qat% poly quaternary = [quote %Quaternary%]
mapextent poly %Catch_Qat%
clipmapextent off
&call SearchBox
clearselect  .ponet poly
reselect     .ponet poly popytype = 2
polygonshade .ponet 57
clearselect  %province% poly
reselect     %province% poly class = 'Country' and name <> 'LESOTHO' and name <> 'SWAZILAND'
polygonshade %province% 32
nselect      %Catch_Qat% poly
polygonshade %Catch_Qat% 81
clearselect  %Catch_Qat% poly
/*
clearselect   %Towns% poly
polygonshade  %Towns% 83
reselect %Catch_Qat% poly quaternary = [quote %Quaternary%]
linesymbol    14
polygons      %Catch_Qat%
linesymbol    4
linecolor     cyan_2
arcs          %RiversOut%

&do nOrder = 1 &to 7
  &sv width = %nOrder% / 70
  &type Rivers of order %nOrder% being drawn with a width of ~
        [calc [round [calc 100 * %width%]] / 100]cm:
  clearselect %Rivers% arc
  reselect    %Rivers% arc order = %nOrder%
  linesize    %width%
  arcs        %Rivers%
&end

clearselect   %Lakes% poly
reselect      %Lakes% poly name <> 'Island'
polygonshade  %Lakes% 56
```

```
overpost      0.1 1.0 0.1
overpost text moveable
overpost on
textalignment automatic

linesymbol 7
arcs .ponet
linesymbol 1
markersymbol %MonPtMrk%
markerscale 2.8
markercolor 7
units  map
marker %mx% %my%
markerscale 2.5
markercolor 4
marker %mx% %my%
units  page

clearselect   %Rivers% arc
reselect      %Rivers% arc mapextent
reselect      %Rivers% arc overlap %Catch_Qat% poly
infofile      %Rivers% arc %BarTmpSrt% length name order init
textsymbol %tsi%
textcolor  blue
reselect %BarTmpSrt% info name <> ''
&sv nRivers = [extract 1 [show select %BarTmpSrt% info]]
&if %nRivers% > 0 &then
&do

  &data ARC INFO
    ARC
    SELECT [translate [entryname %BarTmpSrt%]]
    SORT ON ORDER (D), LENGTH (D)
    Q STOP
  &end

  list %BarTmpSrt% info 1 6
  &sv BigRiver = [show select %BarTmpSrt% info 1 item name]
  &sv BigLength = [show select %BarTmpSrt% info 1 item length]
```

```
  reselect      %Rivers% arc name = [quote %BigRiver%]
  &sv LengthMax = [calc %BigLength% + ( %BigLength% * 0.000001 )]
  &sv LengthMin = [calc %BigLength% - ( %BigLength% * 0.000001 )]
  reselect      %Rivers% arc length > %LengthMin% and length < %LengthMax%
  arctext %Rivers% name # point2 [calc [extract 1 [show textsize]] / 2.5]
&end
textsymbol %tsm%

reselect      %Towns% poly mapextent
reselect      %Towns% poly overlap %Catch_Qat% poly
infofile      %Towns% poly %BarTmpSrt% area name init
reselect %BarTmpSrt% info name <> ''
&sv nTowns = [extract 1 [show select %BarTmpSrt% info]]
&if %nTowns% > 0 &then
&do

  &data ARC INFO
    ARC
    SELECT [translate [entryname %BarTmpSrt%]]
    SORT ON AREA (D)
    Q STOP
  &end

  list %BarTmpSrt% info 1 6
  &sv BigTown = [show select %BarTmpSrt% info 1 item name]
  &sv BigArea = [show select %BarTmpSrt% info 1 item area]
  reselect      %Towns% poly name = [quote %BigTown%]
  &sv AreaMax = [calc %BigArea% + ( %BigArea% * 0.000001 )]
  &sv AreaMin = [calc %BigArea% - ( %BigArea% * 0.000001 )]
  reselect      %Towns% poly area > %AreaMin% and area < %AreaMax%
  labeltext     %Towns% nametag cc
&end

reselect      %Lakes% poly mapextent
reselect      %Lakes% poly overlap %Catch_Qat% poly
reselect      %Lakes% poly name <> 'Island'
reselect      %Lakes% poly name <> '?'
infofile      %Lakes% poly %BarTmpSrt% area name init
textsymbol %tsi%
```

```
textcolor  blue
reselect %BarTmpSrt% info name <> ''
&sv nLakes = [extract 1 [show select %BarTmpSrt% info]]
&if %nLakes% > 0 &then
&do

  &data ARC INFO
    ARC
    SELECT [translate [entryname %BarTmpSrt%]]
    SORT ON AREA (D)
    Q STOP
  &end

  list %BarTmpSrt% info 1 6
  &sv BigLake = [show select %BarTmpSrt% info 1 item name]
  &sv BigArea = [show select %BarTmpSrt% info 1 item area]
  reselect      %Lakes% poly name = [quote %BigLake%]
  &sv AreaMax = [calc %BigArea% + ( %BigArea% * 0.000001 )]
  &sv AreaMin = [calc %BigArea% - ( %BigArea% * 0.000001 )]
  reselect      %Lakes% poly area > %AreaMin% and area < %AreaMax%
  labeltext     %Lakes% name cc
&end

textsymbol %tsm%
overpost   off
MOVE [extract 3 [show maplimits]] [extract 2 [show maplimits]]
textoffset -0.1 0.1
textmask rectangle 0.02
&if %BigFont% &then
&do
  &sv tsc [show textscale]
  TEXTSCALE 2.8
  TEXT %Quaternary% lr
  TEXTSCALE %tsc%
&end
&else
  TEXT %Quaternary% lr

BOX       [show maplimits]
```

```
textoffset 0 0
textmask none
markerSCALE 1
CLIPMAPEXTENT on
&return

/* - - - - - - - - - - - - - - - - - -

&routine PaperPrint

/* routine to delete old plot file and send graph to printer:

&sv plot = %PlotName%

&type rotating %plot% to make %plot%r (Landscape)
&if [exists %plot%r.gra -file] &then &sys %remove% [locase %plot%]r.gra
rotateplot %plot% %plot%r
&type converting %plot%r to %plot%.ps (Postscript)
&if [exists %plot%.ps   -file] &then &sys %remove% [locase %plot%].ps
postscript %plot%r %plot%.ps

&if %Debug% &then
&do
  &type Debug mode: not printing %plot%.ps: print manually if required!
  &type ********************************************
  &type %plot%.ps %plot%r.gra %plot%.gra files kept...
  &type ********************************************
&end
&else
&do
  &if %opsys% = SunOS &then
  &do
    &type Now printing %plot%.ps
    &sys lp -c -d hp8000 [locase %plot%].ps
    &sys lpstat

    &if [exists %plot%.gra  -file] &then &sys %remove% [locase %plot%].gra
    &if [exists %plot%r.gra -file] &then &sys %remove% [locase %plot%]r.gra
    &if [exists %plot%.ps   -file] &then &sys %remove% [locase %plot%].ps
```

```
    &type ********************************************
    &type %plot%.ps %plot%r.gra %plot%.gra files removed
    &type ********************************************
  &end
  &if %opsys% = Windows_NT &then
  &do
    &type %opsys% mode: not printing %plot%.ps: print manually if required!
    &type ********************************************
    &type %plot%.ps %plot%r.gra %plot%.gra files kept...
    &type ********************************************
  &end
&end

&return

/* - - - - - - - - - - - - - - - - - -

&routine PageSet
/* set page constants (A4 only in this version):

&sv xPageSiz = 29.7
&sv yPageSiz = 20.9
&sv xMar     = 0.5
&sv yMar     = 0.5
&sv xPage    = %xPageSiz% - %xMar%
&sv yPage    = %yPageSiz% - %yMar%
&sv Lxmin    = 0.084 * %xPageSiz%
&sv Lymin    = 0.311 * %yPageSiz%
&sv Lxmax    = 0.875 * %xPageSiz%
&sv Lymax    = 0.861 * %yPageSiz%

&return

/* - - - - - - - - - - - - - - - - - -

&routine FlowCalc

&sv nVars = %nVars% + 1
&sv Var%nVars% = Flow
```

```
&sv FlowStn = [substr %Station% 1 6]A01
&if [exists FLOWBARC.TMP -info] &then &sv deli [delete FLOWBARC.TMP -info]
&if [exists %FlowDir%%FlowStn% -file] &then
&do

  &DATA arc info
     ARC
     DFMT YMD-/
     DEFINE FLOWBARC.TMP
     DATE   ,  8,11,D
     FLOW   ,  4, 9,F,3
     CODE   ,  1, 1, C
     NDATE  ,  4,10,F,0

     ADD FROM %FlowDir%%FlowStn%
     CALC NDATE = DATE
  Q STOP
  &end

&end

&else
  &type Sorry - no flow data for station %FlowStn%
&return

/* - - - - - - - - - - - - - - - - -

&routine PlotChem

&if [locase [value Var%n%]] <> flow &then
&do
   &sv yMax = null
   &sv yMin = null
   &if ^ %Debug% &then &message &off
   CLEARSELECT %ChemFile% INFO
   RESELECT    %ChemFile% INFO STATION = [quote %Station%]
   RESELECT    %ChemFile% INFO [value Var%n%] > 0
   RESELECT    %ChemFile% INFO %DateRange%
   INFOFILE %ChemFile% INFO %TempChemFile% date [value Var%n%] init
```

```
&sv ndata = [before [show select %ChemFile% info] ,]
&message &on
&if %ndata% > 0 &then
&do
  &call Median
  &call Percentile
  &sv NoData_AtAll = .FALSE.
  STATISTICS %ChemFile% INFO
  max [value Var%n%]
  min [value Var%n%]
  end

  &sv yMax = [calc [show statistic 1 1] ]
  &sv yMin = [calc [show statistic 2 1] ]
  &if %ndata% = 1 &then &sv yMin = 0
  GRAPHEXTENT %.startday% %yMin% %.endday% %yMax%
  GRAPHBAR %ChemFile% INFO NDATE [value Var%n%] 1

  /* Write the maximum value for variable n on the right-hand y axis.
  /* Do some elementary rounding off...

  &sv var_minva = [calc [round [calc 1000 * %yMin%]   ] / 1000]
  &sv var_medva = [calc [round [calc 1000 * %Median%] ] / 1000]
  &sv var_pctva = [calc [round [calc 1000 * %PcntVar%]] / 1000]
  &sv var_value = [calc [round [calc 1000 * %yMax%]   ] / 1000]

  MOVE [calc %Lxmax% + 0.1] %y2%

  &if %BigFont% &then
  &do
    &sv tsc [show textscale]
    TEXTSCALE 2.5
    &if [locase [value Var%n%]] = conductivity &then
    &do
      TEXTSYMBOL %tss%
      TEXTSTYLE TYPESET
      TEXT [quote %var_value% mSm[unquote '!pat7318;!pat1619;']] ul
      TEXTSYMBOL %tsm%
    &end
```

```
    &else
      &if [locase [value Var%n%]] = ph &then
      &do
        TEXTSYMBOL %tss%
        text [quote %var_value%] ul
        TEXTSYMBOL %tsm%
      &end
      &else
      &do
        TEXTSYMBOL %tss%
        TEXTSTYLE TYPESET
        text [quote %var_value% mg[unquote '!pat7248;!pat7318;!pat1619;']] ul
        TEXTSYMBOL %tsm%
      &end
    TEXTSCALE %tsc%
  &end
  &else
  &do
    &if [locase [value Var%n%]] = conductivity &then
    &do
      TEXTSYMBOL %tss%
      TEXTSTYLE TYPESET
      TEXT [quote (%var_pctva%)%var_value% mSm[unquote '!pat7318;!pat1619;']\%var_minva% (%var_medva%)] ul
      TEXTSYMBOL %tsm%
      TEXTSTYLE TYPESET
    &end
    &else
      text [quote (%var_pctva%) %var_value%\%var_minva% (%var_medva%)] ul

    MOVE [calc %Lxmin% - 0.1] %y1%
    TEXT [quote n=%ndata%] lr

  &end

  &sv ValList = %ValList%,%var_minva%,%var_medva%,%var_pctva%,%var_value%

&end

&if %ndata% = 0 &then
```

```
   &do
     &type No data.
     MOVE [calc %Lxmax% + 0.1] %y2%
     TEXT 'No data' ul
     &sv ValList = %ValList%,-9,-9,-9,-9
   &end
&end
&else
   &type Program error - should not try to plot flow here...

&if ^ %Debug% &then &sv deltemp = [delete %TempChemFile% -info]

&return

/* - - - - - - - - - - - - - - - - - -

&routine PlotFlow

&if [exists FLOWBARC.TMP -info] &then
&do
   CLEARSELECT FLOWBARC.TMP INFO
   RESELECT    FLOWBARC.TMP INFO FLOW >= 0
   RESELECT    FLOWBARC.TMP INFO %DateRange%
   INFOFILE    FLOWBARC.TMP INFO %TempChemFile% date [value Var%n%] init
   &sv ndata = [before [show select FLOWBARC.TMP info] ,]
   &if %ndata% > 0 &then
   &do
     &call Median
     &call Percentile
     STATISTICS FLOWBARC.TMP INFO
     max flow
     end

     &sv Ymax = [calc [show statistic 1 1] ]
     &sv ymin = 0
     GRAPHEXTENT %.startday% %yMin% %.endday% %yMax%
     &if %Ymax% > 0 &then GRAPHBAR FLOWBARC.TMP INFO NDATE flow

     MOVE [calc %Lxmax% + 0.1] %y2%
```

```
      &sv var_minim  = 0
      &sv var_median = [calc [round [calc 1000 * %Median%] ] / 1000]
      &sv var_Pcntl  = [calc [round [calc 1000 * %PcntVar%]] / 1000]
      &sv var_value  = [calc [round [calc 1000 * %yMax%]   ] / 1000]

      &if %BigFont% &then
      &do
        &sv tsc [show textscale]
        TEXTSCALE 2.5
        TEXTSYMBOL %tss%
        TEXTSTYLE typeset
        TEXT [quote %var_value% m[unquote '!pat1621;s!pat7318;!pat1619;']] cl
        TEXTSYMBOL %tsm%
        TEXTSTYLE typeset
        TEXTSCALE %tsc%
      &end
      &else
      &do
        TEXTSYMBOL %tss%
        TEXTSTYLE typeset
        TEXT [quote (%var_Pcntl%) %var_value% m[unquote '!pat1621;s!pat7318;!pat1619;']\(%ymin%)(%var_median%)] cl
        TEXTSYMBOL %tsm%
        TEXTSTYLE typeset
        MOVE [calc %Lxmin% - 0.1] %y1%
        TEXT [quote n=%ndata%] lr

      &end

      &sv ValList = %ValList%,%var_minim%,%var_median%,%var_Pcntl%,%var_value%
    &end
    &else
    &do
       &type No flow data available in %FlowStn% for %DateRange%
       MOVE [calc %Lxmax% + 0.1] %y2%
       TEXT 'No data' ul
       &sv ValList = %ValList%,-9,-9,-9,-9
    &end
&end
&else
```

```
&do
   &type Sorry - no flow data file %FlowStn% in %FlowDir%
   MOVE [calc %Lxmax% + 0.1] %y2%
   TEXT 'No data' ul
   &sv ValList = %ValList%,-9,-9,-9,-9
&end
&return

/* - - - - - - - - - - - - - - - - - -

&routine SetAbbrev

&sv VarAbbr18 = Flow
&sv VarAbbr17 = EC
&sv VarAbbr16 = pH
&sv VarAbbr15 = TDS
&sv VarAbbr14 = Ca
&sv VarAbbr13 = Mg
&sv VarAbbr12 = K
&sv VarAbbr11 = Na
&sv VarAbbr10 = TAL
&sv VarAbbr9  = Cl
&sv VarAbbr8  = F
&sv VarAbbr7  = Si
&sv VarAbbr6  = SO4
&sv VarAbbr5  = NH4(N)
&sv VarAbbr4  = NO3(N)
&sv VarAbbr3  = KN
&sv VarAbbr2  = PO4(P)
&sv VarAbbr1  = TP

&return

/* - - - - - - - - - - - - - - - - - -

&routine AlbGeo

&if [exists %BarScratch% -file] &then &ty [delete %BarScratch% -file]
&if [exists %BarPrjScr% -file]  &then &ty [delete %BarPrjScr% -file]
```

```
&if [exists %BarPrjXY%   -file] &then &ty [delete %BarPrjXY%   -file]
&if [exists %BarPrjGeo%  -file] &then &ty [delete %BarPrjGeo%  -file]
&data arc
  project file %BarTmpXY% %BarScratch%
  input
  projection albers
  units meters
  spheroid clarke1880
  parameters
  -18 0 0
  -32 0 0
  24 0 0
  00 0 0
  0.0
  0.0
  output
  projection geographic
  units dms
  spheroid clarke1880
  parameters
  end
quit
&end

&if %opsys% = SunOS &then &sys nawk 'BEGIN {OFS=","}{print $1,$2,$3,$4,$5,$6}' %BarScratch% > %BarTmpGeo%
&else
  &sys awk "BEGIN {OFS=\",\"}{print $1,$2,$3,$4,$5,$6}" %BarScratch% > %BarTmpGeo%
&sv GeoUnit [open %BarTmpGeo% openstatus -read]
&sv GeoCoord [read %GeoUnit% readstatus]
&sv LonD = [extract 1 %GeoCoord%]
&sv LonM = [extract 2 %GeoCoord%]
&sv LonS = [extract 3 %GeoCoord%]
&sv LatD = [extract 4 %GeoCoord%]
&sv LatM = [extract 5 %GeoCoord%]
&sv LatS = [extract 6 %GeoCoord%]
&sv LonS = [round %LonS%]
&sv LatS = [round %LatS%]
&if %LonS% = 60 &then
&do
```

```
  &sv LonM = %LonM% + 1
  &sv LonS = 0
&end
&if %LonM% = 60 &then
&do
  &sv LonD = %LonD% + 1
  &sv LonM = 0
&end
&if %LonD% > 180 &then &sv LonD = %LonD% - 360
&if %LatS% = 60 &then
&do
  &sv LatM = %LatM% + 1
  &sv LatS = 0
&end
&if %LatM% = 60 &then
&do
  &sv LatD = %LatD% + 1
  &sv LatM = 0
&end
&if %LatD% > 90 &then &sv LatD = %Latd% - 90
&sv LonA = [unquote '']
&sv LatA = [unquote '']
&if %LonD% > 0 &then &sv LonA = E
&if %LonD% < 0 &then &sv LonA = W
&if %LatD% > 0 &then &sv LatA = N
&if %LatD% < 0 &then &sv LatA = S
&if %LonM% = 0 &then &sv LonM = 00
&if %LonS% = 0 &then &sv LonS = 00
&if %LatM% = 0 &then &sv LatM = 00
&if %LatS% = 0 &then &sv LatS = 00
&if %LonM% > 0 and %LonM% < 10 &then &sv LonM = 0%LonM%
&if %LonS% > 0 and %LonS% < 10 &then &sv LonS = 0%LonS%
&if %LatM% > 0 and %LatM% < 10 &then &sv LatM = 0%LatM%
&if %LatS% > 0 and %LatS% < 10 &then &sv LatS = 0%LatS%

/*&sv GeoString = [abs %LatD%]°%LatM%/%LatS%"%LatA% [abs %LonD%]°%LonM%/%LonS%"%LonA%
/*&sv GeoLabel  = [abs %LatD%]°%LatM%/%LatS%"%LatA% [abs %LonD%]°%LonM%/%LonS%"%LonA%
&sv GeoString = [abs %LatD%] %LatM% %LatS% %LatA% [abs %LonD%] %LonM% %LonS% %LonA%
&sv GeoLabel  = [abs %LatD%] %LatM% %LatS% %LatA% [abs %LonD%] %LonM% %LonS% %LonA%
```

```
&return

&routine SearchBox

&sv SrchArea = [show mapextent]
&return

/*-----------------------------------------------------------------------
&routine Median

/* calculate median:

&if %ndata% = 1 &then &sv Median = [show select %ChemFile% info 1 item [value Var%n%]]
&else
  &do
  &data ARC INFO
    ARC
    SELECT [translate [entryname %TempChemFile%]]
    SORT ON [translate [value Var%n%]]
    Q STOP
   q
  &end

  &sv MedRec = [round [max 1 [calc 0.5 * %ndata%] ] ]
  &if %MedRec% > %ndata% &then &sv MedRec = %ndata%
  &if [mod %ndata% 2] = 0 &then    /* even
  &do
    &sv MedRec2 = %MedRec% + 1
    &sv Median1 = [show select %TempChemFile% info %MedRec%  item [value Var%n%]]
    &sv Median2 = [show select %TempChemFile% info %MedRec2% item [value Var%n%]]
    &sv Median  = [calc ( %Median1% + %Median2% ) / 2 ]
  &end
  &else                            /* odd
  &do
    &sv Median  = [show select %TempChemFile% info %MedRec%  item [value Var%n%]]
  &end
&end
```

```
&return


/* commented out:

&sv MapexCat = [show mapextent]

&sv Catx1 = [extract 1 %MapexCat%]
&sv Caty1 = [extract 2 %MapexCat%]
&sv Catx2 = [extract 3 %MapexCat%]
&sv Caty2 = [extract 4 %MapexCat%]

&sv Catxlen = [calc %Catx2% - %Catx1%]
&sv Catylen = [calc %Caty2% - %Caty1%]

/* reduce the search area to half the mapextent:

&sv Srchxlen = [calc %Catxlen% / [sqrt 2]]
&sv Srchylen = [calc %Catylen% / [sqrt 2]]
&sv Srchx    = [calc %Srchxlen% / 2]
&sv Srchy    = [calc %Srchylen% / 2]

&sv Srchx1 = [calc %Catx1% + %Srchx%]
&sv Srchy1 = [calc %Caty1% + %Srchy%]
&sv Srchx2 = [calc %Catx2% - %Srchx%]
&sv Srchy2 = [calc %Caty2% - %Srchy%]
&sv SrchArea = %Srchx1% %Srchy1% %Srchx2% %Srchy2%
&type Mapextent = %MapexCat%
&type SearchArea  = %SrchArea%

&return

/*-------------------------------------------------------------------------
&routine Percentile

/* Calculate nth percentile.  Algorithm supplied by John Carter.
/* Calculate the record, then sort the data for each variable and select the nth percentile.
/* First calculate the (possibly) theoretical ExactRecord for the percentile,
/* then find the two integers surrounding this value:
```

```
&sv PcntRec = [round [max 1 [calc %Percentile% / 100 * %ndata%]]]
&if %PcntRec% > %ndata% &then &sv PcntRec = %ndata%
&sv ExactRec = [calc ( %Percentile% * %ndata% ) / 100 ]
&sv Recd1      [truncate %ExactRec%]
&if %Recd1% < 1 &then &sv Recd1 = 1
&sv Recd2    = [calc %Recd1% + 1]
&if %Recd2% > %ndata% &then &sv Recd2 = %ndata%

&data ARC INFO
  ARC
  SELECT [translate [entryname %TempChemFile%]]
  SORT ON [translate [value Var%n%]]
  Q STOP
 q
&end

&sv FirstValue  = [show select %TempChemFile% INFO %Recd1% item [value Var%n%]]
&sv SecondValue = [show select %TempChemFile% INFO %Recd2% item [value Var%n%]]

/* Interpolate between these two records, by the cunning method of finding out
/* the exact percentile of the two records surrounding the percentile we want:

&sv Val2-1    = [calc %SecondValue% - %FirstValue%]
&sv Pcntile1  = [calc ( %Recd1% / %ndata% ) * 100]
&sv Pcntile2  = [calc ( %Recd2% / %ndata% ) * 100]
&if %Pcntile1% <> %Pcntile2% &then
&do
  &sv PcntRatio = [calc ( %Percentile% - %Pcntile1% ) / ( %Pcntile2% - %Pcntile1% ) ]
  &sv PcntVar   = [calc ( %PcntRatio% * %Val2-1% ) + %FirstValue%]
&end
  &else &sv PcntVar = %FirstValue%

&return

/*-----------------------------------------------------------------------
&routine SetSuffix

/* 1st 2nd 3rd 4th ... 20th 21st 22nd 23rd 24th ... 30th 31st 32nd 33rd .... 100th
```

```
&select %number%
  &when 1,21,31,41,51,61,71,81,91
    &do
      &sv suffix = st
    &end
  &when 2,22,32,42,52,62,72,82,92
    &do
      &sv suffix = nd
    &end
  &when 3,23,33,43,53,63,73,83,93
    &do
      &sv suffix = rd
    &end
  &otherwise
    &do
      &sv suffix = th
    &end
&end

&return

/*-----------------------------------------------------------------------
&routine ISOdate

/* date in ISO format
&sv ISOdate [date -year]-[substr[date -usa] 1 2]-[substr[date -vmsdate] 1 2] - [before[date -vmstime] .]

&return


/*-----------------------------------------------------------------------
&routine SetDefaults

&sv number = %Percentile%; &call SetSuffix
&sv plot dummy

&if [locase %VarType%] = m &then
&do
```

```
  &call SetAbbrev
  &sv Var17 = Conductivity
  &sv Var16 = pH
  &sv Var15 = TDS
  &sv Var14 = Calcium
  &sv Var13 = Magnesium
  &sv Var12 = Potassium
  &sv Var11 = Sodium
  &sv Var10 = TAlkalinity
  &sv Var9  = Chloride
  &sv Var8  = Fluoride
  &sv Var7  = Silica
  &sv Var6  = Sulphate
  &sv Var5  = NH4(N)
  &sv Var4  = NO3(N)
  &sv Var3  = KN
  &sv Var2  = PO4(P)
  &sv Var1  = TP
  &sv nVariables = 17
  &if %inFile% = Default &then &sv ChemFile = %chemdir%wq/inorganic.dat
    &else &sv ChemFile = %inFile%

&end

&if [locase %VarType%] = t &then
&do
  &sv Var21 = Titanium
  &sv Var20 = Antimony
  &sv Var19 = Mercury
  &sv Var18 = Boron
  &sv Var17 = Beryllium
  &sv Var16 = Strontium
  &sv Var15 = Copper
  &sv Var14 = Vanadium
  &sv Var13 = Aluminium
  &sv Var12 = Molybdenum
  &sv Var11 = Chromium
  &sv Var10 = Iron
  &sv Var9  = Manganese
```

```
  &sv Var8  = Nickel
  &sv Var7  = Cadmium
  &sv Var6  = Lead
  &sv Var5  = Zinc
  &sv Var4  = Zirconium
  &sv Var3  = Barium
  &sv Var2  = Cobalt
  &sv Var1  = Arsenic
  &sv nVariables = 21
  &if %inFile% = Default &then &sv ChemFile = %enterprise%/iwqs/prj/aqces/wq_data/trace.dat
    &else &sv ChemFile = %inFile%
&end

/* set file constants:
&if ^ [exists %ChemFile% -info] &then &stop %ChemFile% does not exist!

&sv StnPnt      = %covdir%s-africa/nms_his
&sv Province    = %covdir%s-africa/spr_500
&sv Catch_Pri   = %covdir%s-africa/hca_1
&sv Catch_Sec   = %covdir%s-africa/hca_2
&sv Catch_Ter   = %covdir%s-africa/hca_3
&sv Catch_Qat   = %covdir%s-africa/hca_4
&sv Rivers      = %covdir%s-africa/wri_500
&sv RiversOut   = %covdir%s-africa/wri_not_sa
&sv Lakes       = %covdir%s-africa/wla_500
&sv Towns       = %covdir%s-africa/smu_500
&sv FlowDir     = %enterprise%/iwqs/db/flow/
&sv User        = [translate [substr [username] 1 1]][substr [username] 2 [calc [length [username]] - 1]]
&sv FileName    = bc_[username]
&sv BarTmpLst   = barcodel.tmp
&sv BarTmpSrt   = barcode.tmp
&sv BarTmpXY    = bcdexy.tmp
&sv BarPrjXY    = [before %BarTmpXY% .].prj
&sv BarTmpGeo   = bcdegeo.tmp
&sv BarPrjGeo   = [before %BarTmpGeo% .].prj
&sv BarScratch  = bcdscr.tmp
&sv BarPrjScr   = [before %BarScratch% .].prj
&sv TempChemFile = [unquote bc[substr [username] 1 6].tmp]
```

```
&call ISOdate
&sv RepFileID   = [subst [quote %ISOdate%] ' ' '']
&sv RepFileID   = [subst [quote %RepFileID%] '-' '']
&sv RepFileID   = [unquote [subst [quote %RepFileID%] ':' '']]
&sv Report      = %enterprise%/db/barcode/bc[username]%RepFileID%.txt
&if [exists %Report% -file] &then &sv delf [delete %Report% -file]

&if [exists %enterprise%/db/barcode/barcode.log -file] &then
  &sv LogFile    = %enterprise%/db/barcode/barcode.log
  &else
    &sv Logfile  = %tempdir%/barcode.log

&call DeleteTempFiles
&sv closeall [close -all]
&sv ReportUnit  = [open %Report% openstatus -write]

&type Job run on %ISOdate%
&sv rw = [write %ReportUnit% [quote Data summary on %ISOdate%]]

&sv MonPtMrk = 405
&call PageSet

&return

/*-------------------------------------------------------------------------
&routine CheckDate
/* check and set date limits:

&if %Year1% < 1000 or %Year2% < 1000 &then
  &stop Please use 4-digit years!
&sv Date1 = 1-Jan-%Year1%
&sv ISOdate1 = %Year1%-Jan-1
&sv Date2 = 31-Dec-%Year2%
&sv ISOdate2 = %Year2%-Dec-31
/* Arc8 requires delimiters in dates:
&sv DateRange = Date >= %Year1%-01-01 & date <= %Year2%-12-31
date2day %Date1% .startday
date2day %Date2% .endday
```

```
&sv counter    = 0
&sv ReadStatus = 0
&sv Page       = 1

&return

/*-------------------------------------------------------------------------
&routine DeleteTempFiles

&if [exists %Report%r -file]    &then &sv delf [delete %Report%r -file]
&if [exists %BarTmpXY% -file]   &then &sv delf [delete %BarTmpXY% -file]
&if [exists %BarTmpGeo% -file]  &then &sv delf [delete %BarTmpGeo% -file]
&if [exists %BarPrjXY% -file]   &then &sv delf [delete %BarPrjXY% -file]
&if [exists %BarPrjGeo% -file]  &then &sv delf [delete %BarPrjGeo% -file]
&if [exists %BarScratch% -file] &then &sv delf [delete %BarScratch% -file]
&if [exists %BarPrjScr% -file]  &then &sv delf [delete %BarPrjScr% -file]

&return

/*-------------------------------------------------------------------------
&routine ScreenSaver
/* Save screen to temporary raster file

&if [exists %tempdir%tempras  -file] &then &sv dt [delete %tempdir%tempras  -file]
&if [exists %tempdir%temprasw -file] &then &sv dt [delete %tempdir%temprasw -file]
screensave %tempdir%tempras

&return

/*-------------------------------------------------------------------------
&routine ScreenPrint
/* Convert temporary raster file to JPEG

&type Converting graphics file %PlotName% into JPG...
&sv stn_jpg = %PlotName%.jpg
&sv stn_jgw = %PlotName%.jgw
&if [exists %stn_jpg% -file] &then &sv deljpg [delete %stn_jpg% -file]
&if [exists %stn_jgw% -file] &then &sv deljgw [delete %stn_jgw% -file]
convertimage %tempdir%tempras %stn_jpg% jfif
```

```
&if [exists %stn_jgw% -file] &then &sv deljgw [delete %stn_jgw% -file]

&return


/* - - - - - - - - - - - - - - - - -

&routine PostScriptPrint

/* routine to delete old plot file and save postscript file:

&sv plot = %pdfdir%%PlotName%

&type converting %PlotName% to %plot%.ps (Postscript)
&if [exists %plot%.ps -file] &then &sys %remove% [locase %plot%].ps
postscript %PlotName% %plot%.ps

&if %Debug% &then
&do
  &type Debug mode:
  &type ********************************************************
  &type %plot%.ps %PlotName%.gra files kept...
  &type ********************************************************
&end
&else
&do
  &if %opsys% = SunOS &then
  &do
    &if [exists %PlotName%.gra -file] &then &sys %remove% [locase %plot%].gra
    &type ********************************************************
    &type %PlotName%.gra removed
    &type ********************************************************
  &end
  &if %opsys% = Windows_NT &then
  &do
    &if [exists %PlotName%.gra -file] &then &sys %remove% [locase %plot%].gra
    &type %opsys% mode:
    &type ********************************************************
    &type %PlotName%.gra removed
```

```
    &type %plot%.pdf file will be created when
    &type Acrobat Distiller is active and watching %pdfdir%...
    &type ******************************************************
  &end
&end

&return
```